

CLUBE

Z

80

/83

N.º 13

NESTE NÚMERO

ENCICLOPÉDIA DA LINGUAGEM BASIC (Cont.)	2
INTRODUÇÃO À LINGUAGEM MÁQUINA (Cont.)	3

Programas ZX81

Barreira	4
Batalha Naval	4
Número Secreto	5
Resolução de Equações	6
NOVOS LIVROS	8

Programas SPECTRUM

Beltman	10
Grelha	14
Porta-Aviões	14
Como obter arcos	15
ACCOUNTANT	18
ESPAÇO SPECTRUM (Parte II)	20
SIMULAÇÃO EXECUTADA NO COMPUTADOR	24
NOVOS PROGRAMAS SPECTRUM	25

No Interior:

Cupão de Inscrição

Páginas Centrais:

Esquemas do Circuito Spectrum (Parte I)

É com imenso prazer que vos apresentamos o n.º 13 do CLUBE Z 80

Existem já 12 meses de trabalho atrás deste jornal mensal
que estabelece o contacto entre os utilizadores de microcomputadores,
em especial da SINCLAIR/TIMEX

Tal como nós, outros compradores e utilizadores dos "micros" tem procurado
ideias sobre programas ou conhecem a utilidade de tal ou tal rotina

CLUBE Z 80 é um jornal de utilizadores de microcomputadores,
no qual se publicam programas, exemplos de utilização, artigos pedagógicos, ensaios e ideias...

Possuímos como meta fixa o
ajudar sempre o utilizador de microcomputadores

No caso específico dos distribuidores de microcomputadores, a divulgação deste jornal
(que não possui publicidade nem objectivos lucrativos)
pode ajudar a criar um melhor suporte para os seus clientes

Podemos publicar pequenos anúncios (gratuitos)
sempre que sejam de interesse para os associados

Igualmente, todo o Software ou livros que tenhamos disponíveis
são colocados à disposição dos associados, nas diversas condições que vigorarem na altura

Esperamos a vossa colaboração,
a vossa divulgação, a vossa associação a este esforço
que terá de ser efectivamente existente e actuante

O **CLUBE Z 80** e o seu jornal *dependem de todos vós*

Os Coordenadores

Alexandre Sousa
Joaquim Magalhães
Maria Irene

ENCICLOPÉDIA DA LINGUAGEM BASIC

OUTUBRO / 83

A N D

(Continuação)

CONTINUAMOS NESTE NÚMERO, A ESCREVER SOBRE ESTE OPERADOR LÓGICO. REGRA GERAL, ESTE OPERADOR FORMA UMA CONJUNÇÃO LÓGICA ENTRE DUAS EXPRESSÕES QUE ENVOLVEM OPERADORES CONDICIONAIS.

Se ambas as expressões são verdadeiras, a conjunção é «verdadeira».

Se uma ou ambas são «falsas» a conjunção é «falsa».

O valor numérico de «verdadeiro» é = 1.

O valor numérico de «falso» é = 0.

Todos os valores diferentes de zero são «verdadeiros».

SERÁ CONVENIENTE DISTINGUIR ENTRE:

VALOR LÓGICO e VALOR NUMÉRICO

VALOR LÓGICO é o valor de uma expressão (usando este critério):

valor (diferente de zero) da expressão = «verdadeiro»

valor (nulo) da expressão = «falso»

QUANDO UMA EXPRESSÃO É AVALIADA LOGICAMENTE, SERÁ ATRIBUÍDO UM DE DOIS VALORES NUMÉRICOS:

VERDADEIRO 1
FALSO 0

EXEMPLO:

```
100 IF (A = 10) AND (B <> 3) THEN GOTO 60
200 PRINT (A AND B)
```

SE A = 15 e B = 6 ... então A = verdadeiro B = verdadeiro.

ENTÃO a relação (A AND B) é verdadeira (logicamente) e o número 15 será impresso.

SE A = 16 («verdadeiro») e B = 0 (zero é um valor «lógico» falso).

ENTÃO a relação (A AND B) é falsa e será impresso o valor 0.

A só é falso se = 0

B só é falso se = 0

Tabela de verdade para AND

A	B	A AND B
verd	verd	verd
verd	falso	falso
falso	verd	falso
falso	falso	falso

(Continua no próximo número)

INTRODUÇÃO À LINGUAGEM MÁQUINA

ZX81

Autor: Fernando D'Almeida Preces

(Continuação)

SCROLL (Norte-Sul) + imagem fixa

Extraindo do programa anterior, que continha os 4 movimentos SCROLL, a parte que nos interessa para este ensaio, utilize o código máquina da REM 3, apagando as 2 primeiras REMs. Na 3 apague todos os caracteres dispensáveis, pois inicialmente ela continha 30 caracteres reservados dos quais só serão necessários para este ensaio, 19 instruções máquina.

Modifique a linha 180 para:

```
180 LET B = USR 16 514
```

Acrescente:

```
185 PRINT AT 21, 14; "■ ■ ■ ■"
```

NOTA: A nave é feita com os caracteres gráficos códigos 6, 128 e 134.

Depois RUN 150...

A nave vai aparecer fixa no ecran, enquanto que os caracteres (*) se movem de cima para baixo.

Uma instrução em Basic com uma única linha de caracteres é bastante rápida de modo que o movimento lhes vai parecer normal. Tudo se complica, entretanto, se quisermos (em Basic) aumentar a figura, ocupando mais que uma linha, pois além de uma grande lentidão, a imagem aparece-nos em **fatias** e sobrepondo por causa do movimento SCROLL, as primeiras sobre as últimas linhas.

Vamos a um exemplo, para o demonstrar.

Escreva:

```
184 PRINT AT 19, 15; "A"
185 PRINT AT 20, 14; "■ ■ ■ ■"
186 PRINT AT 21, 15; "V"
```

e outra vez RUN 150...

Como pode verificar, a imagem aparece muito deformada e o SCROLL é lento.

A solução é outra vez o código máquina.

Antes de escrevermos a rotina, vamos fazer um pequeno resumo teórico sobre o seu funcionamento, considerando para o primeiro caso apenas a substituição da linha Basic 185, o que facilitará a compreensão do leitor, pois na rotina completa há mais complexidade.

Um computador envia normalmente para a TV, 25 imagens por segundo e vai cumprindo as instruções na linguagem para o qual foi programado, nos intervalos dessas projecções. Se umas tantas instruções, por exemplo, em Basic, demorarem 1 ou 2 segundos a

ser executadas então os nossos olhos podem distinguir perfeitamente que os caracteres mandados imprimir não são enviados para o ecran todos de uma vez.

Ora em código máquina uma pequena rotina pode ser totalmente executada em ± 500 microssegundos mas mesmo que seja maior e que tenha de ser repartida em 2 ou 3 sequências de imagem ainda assim os nossos olhos estão longe de se aperceberem dessas interrupções e afirmamos normalmente que é de execução instantânea.

Cada imagem a ser formada pelo ZX, é momentaneamente armazenada numa zona da RAM chamada **FICHEIRO DE PROJECCÃO**, que infelizmente não tem um **parque de estacionamento certo**.

A razão desta anomalia, deve-se ao facto do FICHEIRO ser armazenado numa zona imediatamente a seguir à parte da memória reservada à programação em Basic que, como todos já sabemos, depende das dimensões do programa.

Devido a esse vai e vem contínuo da zona ocupada pelo FICHEIRO, o seu primeiro endereço tem de ser memorizado pelo computador que reserva uma das **variáveis do sistema** a que se dá o nome de **D. FILE** para receber essa anotação.

Cada imagem é formada por blocos (entraremos mais tarde em pormenores) de 24 linhas por 32 colunas. Cada coluna é ainda seguida pelo código 118 (NEW LINE).

O FICHEIRO DE PROJECCÃO é, pois, formado por $24 \times 33 = 792$ localizações de memória.

Faça a seguinte experiência, escrevendo:

```
1 000 LET D FILE = PEEK 16 396 +
      256 * PEEK 16 397
```

```
1 010 PRINT AT 8,8; "D. FILE = ";
      D FILE
```

```
1 020 POKE D FILE + 411, 128
      e GOTO 1 000...
```

Verá aparecer a localização anotada na variável D. FILE e um quadrado preto no centro do ecran.

Em seguida substitua na linha 1020 o numero 411, por outro qualquer que esteja compreendido entre 1 e 726, mas tenha em atenção, que os números 33, 66, 99, 132, etc., têm já pelo computador o código 118 (NEW LINE) e se os alterar, descontrola completamente O FICHEIRO que por sua vez altera toda a zona do programa.

As restantes instruções da nossa rotina são preparatórias para o trabalho dum instrução LDIR (já nossa conhecida) que vai enviar para o ecran o número de caracteres que formarão a imagem fixa.

(Continua no próximo número)

JOGOS ZX81/TMS1000

1 K

JOGO 1 — BARREIRA

Neste jogo, você deverá movimentar um «+» no ecran, usando as teclas do cursor para desenhar uma barreira à volta dos «bug» que tem de caçar.

Se o «+» chocar com um «bug», você é morto; mas se a Rainha «bug» (representada por Graphic A) se dirigir para a barreira, ela morrerá e você ganha o jogo.

Atenção! A Rainha «bug» move-se invariavelmente, podendo «saltar» para outra parte do ecran.

Terminado o jogo, accione qualquer tecla para limpar o ecran e começar novo jogo.

```

1 LET X=VAL "9"
2 LET Y=VAL "12"
3 LET A=(RND*9)+5
4 LET B=(RND*8)+4
5 PRINT AT X,Y;"@";AT A,B;"a";AT X,Y;"+"
6 LET X=X+(INKEY$="6")-(INKEY$="7")
7 LET Y=Y+(INKEY$="8")-(INKEY$="5")
8 PRINT AT X,Y;
9 IF PEEK (PEEK 16398+256*PEEK 16399)=CODE "J" THEN GOTO 45
10 PRINT AT A,B;"J"
15 LET A=A+(RND*3)-1.5
20 LET B=B+(RND*3)-1.5
25 IF RND>.9 THEN GOTO 3
30 PRINT AT A,B;
35 IF PEEK (PEEK 16398+256*PEEK 16399)=CODE "+" THEN GOTO 55
40 GOTO 5
45 PRINT AT 4,2;"TU FOSTE APANHADO POR UM ""BUG""
50 GOTO 75
55 FOR N=128 TO 182
60 PRINT AT A,B;CHR$ N
65 NEXT N
70 PRINT "A RAINHA BUG FOI ELETROCUTADA"
75 PAUSE 4E4
80 CLEAR
85 CLS
90 RUN

```

JOGO 2 — BATALHA NAVAL

S — Submarinos (2).

C — Cruzadores (2). Cada cruzador ocupa 2 quadrados.

B — Porta-aviões (1). Ocupa 3 quadrados.

Como fazer o jogo:

P. ex.: Jogador 1 — «S1A S9H C4B C5B C7C C7D B2E B2F B2G» (coloca um submarino na posição 1A e outro em 9H, etc.).

O 2.º jogador tenta localizar a posição dos navios fazendo, por exemplo, «1B». Se não acertar aparece um «+» nessa posição. Se acertar, aparece no ecran uma letra que indica a posição atingida.

Depois de dar entrada do programa, dê entrada das variáveis sem escrever o número de linha. Depois faça GOTO 15. Não use RUN (limparia as variáveis).

```

LET A = 1      LET G = 7
LET B = 2      LET H = 9
LET C = 3      LET I = 10
LET D = 4      LET J = 20
LET E = 5      LET K = J + G
LET F = 6      LET L = H + K

```

```

14 CLS
15 PRINT " ABCDEFGHI"
16 FOR Z=A TO H
17 PRINT AT Z,A;Z
18 NEXT Z
22 DIM A$(K)
25 INPUT A$(A TO K)
30 INPUT B$
35 IF B$="Z" THEN GOTO I+D
40 LET X=VAL B$(A)
50 LET Y=CODE B$(B)
60 PRINT AT X,Y-L;"+"
70 IF X=VAL A$(B) AND Y=CODE A$(C) THEN PRINT AT X,Y-L;A$(A)
80 IF X=VAL A$(E) AND Y=CODE A$(F) THEN PRINT AT X,Y-L;A$(D)
90 IF X=VAL A$(C+E) AND Y=CODE A$(H) THEN PRINT AT X,Y-L;A$(G)
100 IF X=VAL A$(I+A) AND Y=CODE A$(I+B) THEN PRINT AT X,Y-L;A$(I)

110 IF X=VAL A$(I+D) AND Y=CODE A$(I+E) THEN PRINT AT X,Y-L;A$(I+C)
120 IF X=VAL A$(I+G) AND Y=CODE A$(C+F) THEN PRINT AT X,Y-L;A$(I+F)
130 IF X=VAL A$(J) AND Y=CODE A$(J+A) THEN PRINT AT X,Y-L;A$(H+I)
140 IF X=VAL A$(J+C) AND Y=CODE A$(J+D) THEN PRINT AT X,Y-L;A$(J+B)
150 IF X=VAL A$(J+F) AND Y=CODE A$(K) THEN PRINT AT X,Y-L;A$(J+E)

200 GOTO J+I

```

JOGO 3 — NÚMERO SECRETO

Autor: FERNANDO PRECES
Sacavém

```

1 REM PROGRAMA ELABORADO POR ALMEIDA PRECES, EM 5/4/82.
2 REM "8"
5 PRINT ,,,, " NUMERO SECRETO"
7 PRINT ,,"UM NUMERO AO ACASO SERA GERADO"
8 PRINT ,,"ENTRE 0 E 50."
10 PRINT ,,"VOCE DARA UM NUMERO PALPITE"
12 PRINT ,,"QUE SERA O LIMITE SUPERIOR OU"
13 PRINT ,,"INFERIOR NO QUADRADO APRESENTADO."
15 PRINT ,," SE NAO ACERTOU VOLTE A TENTAR."
25 PAUSE 600
30 CLS
40 LET A=0
42 LET B=50
45 LET C=INT (RND*50)+1
46 FOR D=1 TO 20
50 PRINT AT 6,6;"NUMERO SECRETO"
55 PRINT AT 10,4;" ENTRE ";A;" E ";B
60 INPUT E
65 PRINT AT 15,6;"PALPITE ";D;" **";E;"**"
70 IF E=C THEN GOTO 85
75 IF E>C THEN LET B=E
76 IF E<C THEN LET A=E
80 NEXT D
81 IF D>20 THEN GOTO 30
85 PRINT AT 17,0;"A C E R T O U...NA JOGADA ";D
86 PAUSE 300
87 GOTO 30

```


RESOLUÇÃO DE EQUAÇÕES

ZX81

Suponha uma equação da forma $f(x) = 0$, como por exemplo:

$$y = f(x) = x^3 + 3x^2 - 10x - 5$$

As raízes desta equação são os valores de x para os quais $y = 0$. Pelo gráfico da figura 1 pode verificar que a curva corta o eixo do x em três pontos: o primeiro entre $x = -5$ e $x = -4$; o segundo entre $x = -1$ e $x = 0$ e o terceiro entre $x = 2$ e $x = 3$. Pode obter estas mesmas informações fazendo uma tabela de valores para o intervalo $(-10, 10)$ — fig. 2. Onde y troca de sinal (mais por menos ou menos por mais) é o intervalo no qual pode encontrar uma raiz. A tabela apresenta então os três intervalos: $(-5, -4)$, $(-1, 0)$ e $(2, 3)$ que necessitam ser estudados mais detalhadamente.

Pelo método de bissecção cada intervalo é dividido em duas partes iguais, decidindo a parte onde se encontra a raiz, a qual é novamente dividida, repetindo-se o processo.

- Quanto maior for o intervalo, mais probabilidades tem, de encontrar todas as raízes da equação.
- Não será possível encontrar as raízes para as equações em que o gráfico não corta o eixo do x : por exemplo — $y = x^2 + 1$ (fig. 3).

Como usar o programa

Para melhor explicar a utilização do programa vamos formar a equação $x^3 + 3x^2 - 10x - 5$ como exemplo.

OPÇÃO 1 — Dê entrada da equação do seguinte modo:

$$x * x * x + 3 * x * x - 10 * x - 5$$

Na OPÇÃO 2 escolha o intervalo $(-10, 10)$, seguindo na OPÇÃO 3 os vinte pontos do intervalo como é lógico.

A OPÇÃO 4 apresentará a tabela de valores tal como na fig. 2. Pode então verificar os intervalos onde se dá a troca de sinal. Regresse às opções e na 2 dê entrada de um dos intervalos — ex.: $(-5, -4)$. Propomos o erro 0.00001 (OPÇÃO 5) e pela OPÇÃO 6 verifique a raiz no intervalo escolhido $(-4.84946...)$.

Pode de novo regressar à OPÇÃO 2 e procurar as raízes nos outros 2 intervalos.

Note que a equação utilizada como exemplo só poderia ter 3 raízes.

São possíveis também equações trigonométricas tais como: $\sin(x) + \tan(x) - 3x^2 + 4x$.

A OPÇÃO 7 fará a listagem de todos os parâmetros utilizados e finalmente a OPÇÃO 8 pára o programa.




Fig. 1

X=-10	F(X)=-605
X=-9	F(X)=-401
X=-8	F(X)=-245
X=-7	F(X)=-131
X=-6	F(X)=-53
X=-5	F(X)=-5
X=-4	F(X)=19
X=-3	F(X)=25
X=-2	F(X)=19
X=-1	F(X)=7
X=0	F(X)=-5
X=1	F(X)=-11
X=2	F(X)=-5
X=3	F(X)=19
X=4	F(X)=67
X=5	F(X)=145
X=6	F(X)=259
X=7	F(X)=415
X=8	F(X)=619
X=9	F(X)=877
X=10	F(X)=1195

Fig. 2

Tabela de valores entre -10 e 10 de $x * x * x + 3 * x * x - 10 * x - 5$




Fig. 3

RESOLUÇÃO DE EQUAÇÕES

ZX₈₁

ZX COMPUTING AGO./SET. 83

```
1 SLOW
20 REM (A,B)
30 LET A=-1
40 LET B=1
60 LET A$="X"
80 LET E=0.001
100 LET N=10
110 LET X=0
200 PRINT AT 0,10;"opcoes"
210 PRINT AT 2,0;"1- EXPRESSAO DE F(X)"
220 PRINT AT 4,0;"2- INTERVALO PARA O QUAL DESEJA      ENCONTRAR A RAIZ"

230 PRINT AT 7,0;"3- NUMERO DE PONTOS"
240 PRINT AT 10,0;"4- TABELA DE VALORES DEFINIDA      PELO INTERVALO (A,B)"

250 PRINT AT 13,0;"5- ERRO APROXIMADO"
260 PRINT AT 15,0;"6- RAIZ NO INTERVALO (A,B) PELO    METODO DE BISSECCAO"

270 PRINT AT 18,0;"7- LISTAGEM DE PARAMETROS"
280 PRINT AT 20,0;"8- STOP"
300 LET Z$=INKEY$
310 IF Z$="" OR CODE Z$<29 OR CODE Z$>36 THEN GOTO 300
320 LET Z=VAL Z$
330 CLS
340 GOSUB 1000*Z
350 CLS
360 GOTO 200
1000 PRINT AT 5,0;"EXPRESSAO DE F(X)"
1010 INPUT A$
1020 PRINT AT 8,0;"F(X)=";A$
1030 GOTO 9000
2000 PRINT AT 5,0;"ESPECIFIQUE O INTERVALO      (A=?,B=?)"
2010 PRINT AT 9,2;"A="
2020 INPUT A
2030 PRINT AT 9,6;A
2040 PRINT AT 10,2;"B="
2050 INPUT B
2060 IF B<=A THEN GOTO 2050
2070 PRINT AT 10,6;B
2080 GOTO 9000
3000 PRINT AT 5,0;"NUMERO DE PONTOS"
3010 INPUT N
3020 IF N<>INT N OR N<1 THEN GOTO 3010
3030 PRINT AT 8,0;"N=";N
3040 GOTO 9000
4000 FOR X=A TO B STEP (B-A)/N
4010 PRINT "X=";X;TAB 15;"F(X)=";VAL A$
4020 IF PEEK 16442>4 THEN GOTO 4080
4030 PRINT "PRESS C PARA CONTINUAR"
4040 LET Z$=INKEY$
4050 IF Z$="" THEN GOTO 4040
4060 IF Z$<>"C" THEN GOTO 8990
4070 CLS
6000 LET A1=A
6010~LET B1=B
6020 LET X=A
6030 LET VA=VAL A$
6040 LET X=B
6050 LET VB=VAL A$
6060 IF VA*VB<0 THEN GOTO 6090
```

```

6070 PRINT AT 5,0;"NAO EXISTEM RAIZES NO INTERVALO (<";A;"<";B;">";AT 8,0;"F(X) D
EVE TER SINAL DIFERENTE EM A E B"
6080 GOTO 9000
6090 PRINT AT 2,6;"ESPERE UM MOMENTO"
6100 LET X=(A1+B1)/2
6110 LET ME=(B1-A1)/2
6120 IF ME<E THEN GOTO 6210
6130 LET VX=VAL A$
6140 LET STORE=X
6150 LET X=A1
6160 LET VA=VAL A$
6170 LET X=STORE
6180 IF VA*VX<=0 THEN LET B1=X
6190 IF VA*VX>0 THEN LET A1=X
6200 GOTO 6100
6210 PRINT AT 5,0;"O VALOR DA RAIZ E ";X
6220 PRINT AT 7,0;"ERRO MAXIMO E ";E
6230 GOTO 9000
7000 PRINT AT 5,0;"A EXPRESSAO"
7010 PRINT AT 6,2;"F(X)=";A$
7020 PRINT AT 8,0;"O INTERVALO E (<";A;"<";B;">"
7030 PRINT AT 10,0;"O NUMERO DE PONTOS E ";N
7040 PRINT AT 13,0;"O ERRO MAXIMO PARA A SOLUCAO ";E
7050 GOTO 9000
8000 STOP
8990 IF INKEY$("<") THEN GOTO 8990
9000 PRINT AT 19,0;"TECLA C- IMPRESSORA QUALQUER TECLA PARA REGRESSA
R OPCOES"
9010 LET Y$=INKEY$
9020 IF Y$="" THEN GOTO 9010
9030 IF Y$("<")="C" THEN RETURN
9040 REM 70 ESPACOS NA PROXIMA LINHA
9050 PRINT AT 19,0;"
"
9060 COPY
9070 GOTO 9000
9080 SAVE "F(X)"

```

NOVOS LIVROS

● SPECTRUM GRAPHICS

HAMPSHIRE NICK, DUCKWORTH, LONDON

INICIAÇÃO ÀS TÉCNICAS DE PROGRAMAÇÃO NECESSÁRIAS AO TRAÇADO DE GRÁFICOS NUM SPECTRUM:

- Gráficos de alta resolução
- Gráficos «Plotting»
- Memória Vídeo
- Escalas e Variações
- Rotação e Movimento
- 3 D Displays

PREÇO: (Fotocópias) — 385\$00

● GAMES FOR YOUR ZX SPECTRUM

SHAW PETER, INTERFACE/VIRGIN BOOKS, 1983

MAIS DE 20 PROGRAMAS ENTRE OS QUAIS:

- «Excalibur against the enemy»
- «Ascot»
- «Touchdown on Mars»
- «Munchie Man»

ESTE LIVRO CONTÉM AS INSTRUÇÕES NECESSÁRIAS À INTRODUÇÃO DOS PROGRAMAS NA MÁQUINA, INCLUINDO 1 DICIONÁRIO DE TERMOS DE INFORMÁTICA E AINDA 1 LISTA BIBLIOGRÁFICA SELECIONADA.

PREÇO: (Fotocópias) — 260\$00

- **THE WORKING SPECTRUM,**
LAWRENCE DAVID,

VOLUME 1, SUNSHINE BOOKS, LONDON

Baseado numa colecção de programas sofisticados na área de finanças, cálculo, gráficos, armazenamento de dados, educação, etc. Cada programa é detalhadamente explicado, linha a linha, contendo subrotinas e módulos que, compreendidos, podem servir de base a outros programas que queira fazer.

PREÇO: (Fotocópias) — 450\$00

- **40 BEST MACHINE CODE ROUTINES FOR THE ZX SPECTRUM**

HARDMAN JOHN E HEWSON ANDREW, HEWSON CONSULTANTS, GREAT BRITAIN, 1982

- A estrutura interna do ZX Spectrum
- A linguagem máquina do Z80 - Byts, Bytes, endereços, Glossário de instruções em código máquina
- Rotinas (SCROLL, DISPLAY...)

PREÇO: (Fotocópias) — 310\$00

- **LA PRATIQUE DU ZX SPECTRUM (2 VOLUMES)**

BELLEFONDS, XAVIER LINANT, P.S.I., FRANÇA, 1983

Tomo I:

- Aprofundamento do Basic
- Iniciação à Linguagem Máquina
- Tratamentos de cadeias e de quadros
- Gráficos de alta resolução e utilização das funções matemáticas.
- A cor, o som e a memória do SPECTRUM

PREÇO: (Fotocópias) — 340\$00

Tomo II:

- Programação em Linguagem Máquina

PREÇO: (Fotocópias) — 340\$00

- **THE SPECTRUM HANDBOOK**

LANGDELL TIM, CENTURY PUBLISHING, LONDON, 1982

- Iniciar c/ um ZX SPECTRUM:
Como funciona, a cor e o som:
Programação BASIC: gráficos, jogos, aplicações didáticas e científicas.
- O aproveitamento máximo do Spectrum
- Código Máquina.

PREÇO: (Fotocópias) — 460\$00

- **INITIATION A LA PROGRAMMATION PAR LE BASIC**

SCHMIT JACQUES, EYROLLES, PARIS, 1982

- Estrutura de um sistema informático
- Princípios de funcionamento de um computador
- Introdução ao BASIC
- Noção de programa — Aplicações, exercícios
- Cadeias de caracteres
- Vectores, quadros, variáveis, rotinas, dados, instruções complementares ...
- Utilização do PRINT e LPRINT com USING
- Funções aritméticas

PREÇO: (Fotocópias) — 1 050\$00

- **BASIC HYDRAULICS**

PD SMITH, BUTTERWORTHS, ENGLAND, 1982

- Introdução ao Basic
- Elementos de Fluidos mecânicos
- Engenharia hidráulica, processamento de dados ...

PREÇO: (Fotocópias) — 350\$00

BELTMAN

SPECTRUM

OBJECTIVO DO JOGO

UM ELECTRICISTA TERÁ DE VERIFICAR TODAS AS CAIXAS COM FUSÍVEIS, EXISTENTES NA FÁBRICA. PARA ISSO VAI MOVER-SE DESDE O PONTO INFERIOR AO SUPERIOR, USANDO AS TECLAS 5 a 8. Portanto, vai ter de ser mover rapidamente

DEVERÁ TENTAR ATINGIR O PONTO MAIS ALTO (NÍVEL — 21) QUE SE ENCONTRA NUMERADO E CONFORME O PONTO DE CHEGADA, SERÃO ATRIBUÍDOS BÓNUS (de 10 a 300 pontos, da esquerda para a direita).

Para isso, deve passar através das aberturas de cada nível, evitando, no entanto, as de nível inferior.

A melhor forma de evitar as aberturas de nível inferior, será utilizar as caixas para descansar e ao mesmo tempo esperar que as aberturas desapareçam debaixo do lugar onde se encontra.

INICIALMENTE:

Usar o programa da Pág. 11 (canto inferior direito).
ACCIONE O COMANDO: RUN.

SEGUIDAMENTE: Dar entrada do primeiro endereço da primeira secção (23 760 a 23 786); dar entrada dos 27 bytes.

ACCIONE O COMANDO: RUN.

SEGUIDAMENTE: Dar entrada do primeiro endereço da segunda secção (23 800 a 23 826); dar entrada dos 27 bytes.

ACCIONE O COMANDO: RUN.

SEGUIDAMENTE: Dar entrada do primeiro endereço da terceira secção (23 846 a 23 868); dar entrada dos últimos 23 bytes.

DEPOIS DE TER DADO ENTRADA DAS TRÊS SECÇÕES DO CÓDIGO MÁQUINA, PODE DAR INÍCIO À INTRODUÇÃO DO PROGRAMA A PARTIR DA LINHA 3.

NOTA: NAS LINHAS 20, 45, 75: USE GRAPHICS e tecla A
NAS LINHAS 46, 76 : USE GRAPHICS e tecla B

```

1 REM *VAL \E2VAL \ RESTORE MOVE # GO SUB VAL :VAL \<xxxxxxx>#VAL \E2V
AL \ RESTORE MOVE + GO SUB LN :VAL \<xxxxxxx>
2 REM xxxxxx>G PRINT STEP DRAW \ LET AND G PRINT STEP CLOSE #\ LET AND
RETURN PI>= FOR xx
4 GO SUB 600: LET sc=0
5 PAPER 5: BORDER 5: INK 0: CLS
6 GO SUB 1000
7 FOR a=0 TO 31: PRINT AT 0,a;(a-INT (a/10)*10);" " NEXT a
10 LET r=21: LET c=1
11 DATA 0,16,80,60,18,16,40,32
12 DATA 0,16,18,60,80,16,40,8
13 DATA 0,16,80,60,18,16,40,32
18 LET q=0
20 PRINT AT 21,1;" "
25 FOR t=1000 TO 0 STEP -2
29 PRINT #0;AT 1,0;"Score>=";sc;" " ;TAB 13;"Time>=";t;" "
30 LET n=(INKEY$="8")-(INKEY$="5")
40 IF c+n>30 OR c+n<1 THEN GO TO 60
43 IF n=0 THEN GO TO 60
44 LET q=NOT q
45 PRINT AT r,c; OVER 1;" "
46 PRINT AT r,c+n; OVER 1;" "
50 LET c=c+n
60 LET m=(INKEY$="6")-(INKEY$="7")
70 IF r+m>21 OR r+m<0 THEN GO TO 100
72 IF POINT (c*8+2,175-r*8) AND m=-1 THEN GO TO 100
73 IF m=0 THEN GO TO 100
74 LET q=NOT q
75 PRINT AT r,c; OVER 1;" "
76 PRINT AT r+m,c; OVER 1;" "
77 LET sc=sc-10*(m)
80 LET r=r+m
90 IF r=0 THEN LET sc=sc+10*c: GO TO 2000
100 FOR g=0 TO 30-sc/80 NEXT g
101 IF INKEY$<>" " THEN BEEP .01,21-r
300 LET l=USR 23846

```



```

302 POKE 23675,w+q*8
303 IF SCREEN$(r+1,c)=" " THEN LET m=1: GO TO 74
305 NEXT t
306 GO TO 3000
600 FOR k=USR "a" TO USR "d"-1: READ a: POKE k,a: NEXT k
630 LET w=PEEK 23675
660 RETURN
1002 OVER 0
1005 LET l=1
1007 PRINT #0;AT 0,0; INK 5;"-----"
1010 FOR c=7 TO 159 STEP 8
1011 LET l=NOT l: IF l THEN GO TO 1020
1015 PLOT 0,c: DRAW 0,8,PI
1016 PLOT 255,c: DRAW 0,8,-PI
1020 PLOT 0,c: DRAW 255,0
1021 FOR d=1 TO 2+sc/500
1023 PRINT AT 21-INT (c/8),RND*29+1;" ": NEXT d
1027 NEXT c
1029 OVER 1: PAPER 8
1030 FOR k=1 TO 21 STEP 2: PRINT AT k,0: PAPER 6;"
": NEXT k
1050 FOR k=0 TO 10: PRINT AT RND*20+1,RND*29+1;"#": NEXT k
1070 RETURN
2000 FLASH 1: PAPER 4: CLS
2020 PRINT AT 10,8;"BONUS ";c*10;" pontos";AT 15,8;"somou ";sc;" pontos"
2050 FOR c=1 TO 50: BEEP .01,21: BEEP .01,26: BEEP .03,.31: NEXT c: FLASH 0: RUN

3000 PAPER 0: INK 7: BORDER 0: CLS
3020 PRINT AT 17,8;"Conseguiu ";sc;" pontos."
3030 INPUT "Novo jogo ? ";y$: IF y$="s" OR y$="S" THEN RUN
5000 FOR c=0 TO 7777: LET l=1+(INKEY$(">")): PRINT AT 0,0;l: NEXT c

```

23760 = 42	23800 = 42	23846 = 62
23761 = 176	23801 = 176	23847 = 7
23762 = 92	23802 = 92	23848 = 71
23763 = 69	23803 = 69	23849 = 245
23764 = 14	23804 = 14	23850 = 205
23765 = 0	23805 = 255	23851 = 252
23766 = 205	23806 = 205	23852 = 92
23767 = 170	23807 = 170	23853 = 241
23768 = 34	23808 = 34	23854 = 198
23769 = 126	23809 = 126	23855 = 8
23770 = 50	23810 = 50	23856 = 71
23771 = 176	23811 = 176	23857 = 245
23772 = 92	23812 = 92	23858 = 205
23773 = 229	23813 = 229	23859 = 212
23774 = 209	23814 = 209	23860 = 92
23775 = 35	23815 = 43	23861 = 241
23776 = 1	23816 = 1	23862 = 198
23777 = 31	23817 = 31	23863 = 8
23778 = 0	23818 = 0	23864 = 254
23779 = 237	23819 = 237	23865 = 167
23780 = 176	23820 = 184	23866 = 200
23781 = 58	23821 = 58	23867 = 24
23782 = 176	23822 = 176	23868 = 235

23783 = 92	23823 = 92
23784 = 18	23824 = 18
23785 = 0	23825 = 0
23786 = 201	23826 = 201

BELTMAN: As três secções de código máquina que observámos ao lado, são introduzidas com este programa:

```

1 REM (75 X's) (escrever a letra X,
75 vezes)
2 REM (30 x's)
3 INPUT A,B: FOR N = A TO B: IN-
PUT (N);
": X: POKE N,X: PRINT N,X: NEXT
N.

```

Sinclair

ZX Spectrum Circuit Diagram - Issue Two

SRC 102


© 16 August 1982


Sinclair Research Ltd.

Drawn by J.G. Mathieson


J.G. Mathieson

Nota: A parte 2 sai no próximo número





A14	A15	1	SLOT
A12	A13	2	DO
+5V	D7	3	D1
+9V		4	D2
		5	D6
		6	D5
		7	D3
		8	D4
		9	INT
		10	NMI
		11	HMT
		12	MREQ
		13	IOREQ
		14	PB
		15	UR
		16	-5V
		17	WRIT
		18	+12V
		19	+12V
		20	-12V
		21	M
		22	AP5M
		23	A8
		24	A10
		25	A11



GRELHA

SPECTRUM — 16 ou 48 K

PROGRAMA ADAPTADO P/JOSE MARTINS

```

1 PRINT AT 4,0;"programa adaptado p/JOSE MARTINS";AT 6,0;"SPECTRUM 16K OU 48K";AT 8,0;"A GRELHA DESENHADA SERVE PARA VERIFICAR A DEFINICAO DO TV E EVENTUALMENTE CORRIGIR O SINCRONISMO VERTICAL HORIZONTAL"
2 PAUSE 400:CLS
5 REM "GRELHA"
10 FOR N=0 TO 150 STEP 15
20 FOR M=0 TO 150
30 PLOT N,M
35 PLOT M,M
40 PLOT M,N
45 PLOT M,150-M
50 NEXT M
60 NEXT N
70 REM "COPYRIGHT 04/1983"

```

PORTA-AVIÕES

SPECTRUM — 16 ou 48 K

PROGRAMA ADAPTADO P/JOSE MARTINS

SPECTRUM 16/48 K

OBJECTIVO:

um aviao pretende aterrar num
Porta-Aviões
usar as teclas 6 e 7

```

1 PRINT "PROGRAMA ADAPTADO P/JOSE MARTINS";AT 4,0;"SPECTRUM 16/48 K";AT 6,0;"OBJECTIVO:";AT 8,0;"um aviao pretende aterrar num Porta-Aviões";AT 10,0;"usar as teclas 6 e 7":PAUSE 400:CLS
5 REM "PAV"
10 LET X=10
20 LET Y=X
33 FOR S=0 TO 26
50 CLS
55 PRINT AT 20,0;"-----"
60 PRINT AT 20,S;"
70 IF S=26 THEN LET S=0
80 PRINT AT 20,S;"
150 LET Y=Y+2.5
155 IF Y>27 THEN LET Y=0
160 LET X=X+(INKEY$="6")-(INKEY$="7")
200 IF X=19 AND (Y+3=S+2 OR Y+3=S+3 OR Y+3=S+4) THEN PRINT "BRAVO":STOP
250 IF X=19 AND (Y+3=S OR Y+3=S+1 OR Y+3=S+5 OR Y+3=S+6) THEN GO TO 500
300 IF X>20 THEN GO TO 550
400 NEXT S
500 PRINT AT X,Y;"CHOQUE"
510 STOP
550 PRINT AT X,Y:"PI DE"

```



60 Print at 20, S: "desenho à esquerda"

100 Print at X, Y: "desenho do Porta-Aviões"; at X+1, Y+3: "O"

INSTANT IMAGE TRANSFER TO ZX SPECTRUM RD DIGITAL TRACER £55.50



RD DIGITAL TRACER ZX SPECTRUM E ZX81 para transferir desenhos do papel para o ecran, com software e manual. Vendo ou troco por impressora zx sinclair. Vendo também RS 232 C INTERFACE adaptável a várias impressoras da "COBRA TECHNOLOGY L." para o SPECTRUM (LLIST & LPRINT).

Telef. 61373 • SANTIAGO RIBAS • PORTO

The high resolution colour graphics of the ZX Spectrum permit accurate presentation of complex or irregular images – maps, technical drawings, even personalities. But entering individual co-ordinates for unusual shapes can be tedious and time-consuming.

The RD DIGITAL TRACER cuts out tedious plotting. It provides instant transfer from original to display file – for screen display, ZX printer printout, or retention on cassette.

The RD DIGITAL TRACER is supplied with tracing sheet, software cassette and full instructions on use. Latest software includes fast colour fill, audible keystroke acknowledgement, fast PAPER colour change and fast LOAD from saved display file.

This is an invaluable tool for engineers, architects and other technicians, and for educational use in the presentation of lectures, and in computer training and application. Designed for the ZX Spectrum, the RD DIGITAL TRACER as supplied is compatible with ZX 81, although high resolution colour graphics are not available on this machine.

The RD DIGITAL TRACER is available from computer shops or direct from RD Laboratories. The direct, UK only, price of £55.50 includes VAT (Postage and Packing free). Send a cheque (payable to RD Laboratories Ltd.) with order for delivery within 28 days, or ask your local dealer for details.

COMO OBTER ARCOS?

SPECTRUM

Para representar um arco tal como na figura 1 basta introduzir a seguinte linha:

```
10 PLOT 120,55:DRAW 0,25,9999 — 2 * PI * INT
(9999/(2 * PI))
```

«PLOT» — traduz a posição do arco no ecran segundo as coordenadas x,y.

«DRAW» — define o arco.

Alterando a segunda variável da instrução DRAW para: «...DRAW 0,50...» vai permitir-lhe o segundo arco representado na figura: «...DRAW 0,100...» permite-lhe o terceiro.

Na figura 2 a alteração é a seguinte: «...DRAW 0,-25...» (primeiro arco); 0,-50 (segundo arco); 0,-100 (terceiro arco).

Fig. 3 — As variáveis do comando DRAW, tomam os valores: -25,0; -50,0; -100,0 para primeiro, segundo e terceiro arcos, respectivamente.



Fig. 1



Fig. 2




Fig. 3

NOTA: Dê à instrução «PLOT» os valores que lhe tornem possível o arco completo nos limites do ecran. Caso contrário verificará o erro: «B Integer out of range...».

A linha não deve sofrer qualquer alteração excepto nos comandos DRAW e PLOT (x,y).

Os valores das variáveis podem ser substituídos de modo a definir o arco como pretender.


Fig. 4



```

10 FOR n=68 TO 173 STEP 15
20 READ z: INK z
30 PLOT n,-4
40 DRAW 50,87,-PI/3
50 DRAW 4,-7,-PI
60 DRAW -100,0,-PI/3
70 DRAW 4,7,-PI
80 DRAW 50,-87,-PI/3
90 DRAW -0,0,-PI
100 NEXT n
110 DATA 1,2,3,4,4,3,2,1


```



```

30 LET a=-.508*PI
40 PLOT 120,140
50 DRAW -32,-120,a
60 DRAW 88,88,a
70 DRAW -120,-32,a
80 DRAW 120,-32,a
90 DRAW -88,88,a
100 DRAW 32,-120,a
110 DRAW 32,120,a
120 DRAW -88,-88,a
130 DRAW 120,32,a
140 DRAW -120,32,a
150 DRAW 88,-88,a
160 DRAW -32,120,a


```




```

10 FOR n=0 TO 174
20 PLOT n,0: PLOT n,175
30 NEXT n
40 FOR n=0 TO 174
50 PLOT 0,n: PLOT 175,n
60 NEXT n
70 LET a=0
80 FOR n=167 TO 0 STEP -8
90 LET a=a+8
100 PLOT 0,n: DRAW a,-n
110 DRAW n,a
120 DRAW -a,n
130 DRAW -n,-a
140 NEXT n

```



Alterando a linha 80 do terceiro programa para: TO 0 STEP -16
Obtemos este nova apresentacao.



Alt.: 79
Larg.: 79
n: 2
e: 9

```

10 INPUT "Largura ";a
20 INPUT "Altura ";b
30 INPUT "Ent. n ";n
40 INPUT "Ent. e ";e
50 PLOT 0,87: DRAW 175,0
60 PLOT 87,0: DRAW 0,175
70 LET c=.001: LET d=2*PI
90 FOR o=c TO d STEP .001
100 LET x=a*SIN (n*o+e)+87: LET y=b*SIN o+87
110 PLOT x,y
120 NEXT o
130 PRINT AT 2,22;"Alt.: ";b;AT 4,22;"Larg.: ";a;AT 6,22;"n: ";n;AT 8,22;"e: "
;e
140 INPUT " Impressora:I  Fim:F  Repet.:R  CLS:C "; LINE a$
150 IF a$="i" THEN COPY
160 IF a$="f" THEN STOP
170 IF a$="r" THEN GO TO 10
180 IF a$="c" THEN RUN

```

CURVA DE LISSAJOUS

Este programa desenha uma curva de Lissajous com uma altura e largura especificadas, para um dado número de ciclos e com uma certa simetria.

(A execução do gráfico é bastante lenta).

LISTA DAS VARIÁVEIS

- a — largura da figura
- b — altura da figura
- n — número de ciclos
- e — simetria
- o — deslocação angular em radianos
- x e y — coordenadas do PLOT
- a\$ — variável decisão

Para aquisição de um ZX Spectrum
VENDO PELA MELHOR OFERTA:

- ZX 81
- impressora Sinclair
- módulo de expansão de 32 Kbytes
- Circuito de inversão de vídeo e sonorização do teclado
- várias cassetes com cerca de 90 programas (utilitários e jogos)
- literatura para ZX 81 (livros e revistas)

Contactos: A. OLIVEIRA

Rua da Alegria, 726-3.ª-A — 4000 PORTO
Telefone 569181

OBSERVAÇÃO

JOGO DO GALO

(número de Setembro)

No Jogo do Galo para o ZX₈₁, cuja listagem saiu no número anterior (n.º 12, Set./83), na pág. 8, as linhas 150 e 240 (LET T = CODE ".....") são completas com as teclas correspondentes ao código dos gráficos que se encontram nas linhas 500 e 510. Para isso use a tecla Graphics +, a tecla indicada na REM (500, 510).

ACCOUNTANT

SPECTRUM 48 K

Autor: SANTIAGO RIBAS
Porto


Trata-se de uma rotina importante para o utilizador do SPECTRUM.

Faça o «MERGE» nesta rotina com o seu programa (que não contenha nenhuma linha coincidente!) e poderá ter o desenho do diagrama de fluxo do seu programa.

```

9900 REM "flow-chart"
9901 OVER 0: INVERSE 0: BRIGHT 1: BORDER 1: PAPER 6: INK 0: CLS
9902 LET a=PEEK 23636: LET a=a*256: LET a=a+PEEK 23635
9903 LET c=0: LET y=175: LET x=130
9904 LET l=PEEK a: LET l=l*256: LET l=l+PEEK (a+1): LET a=a+4
9905 IF l>9899 THEN GO TO 9943
9906 PRINT l;TAB 5;CHR$ PEEK a: LET c=c+2: PRINT
9907 PLOT x,y: LET b=PEEK a
9908 IF b=226 OR b=234 OR b=242 OR b=254 THEN GO TO 9924
9909 IF b>227 AND b<231 OR b=232 OR b=235 OR b=241 OR b=247 OR b=249 OR b=253 TH EN GO TO 9925
9910 IF b=243 OR b=250 THEN GO TO 9926
9911 IF b=236 OR b=237 THEN GO TO 9934
9912 DRAW 16,0: DRAW -4,-8: DRAW -28,0: DRAW 4,8: DRAW 12,0
9913 LET y=y-8
9914 IF b=226 OR b=236 OR b=237 OR b=254 THEN GO TO 9916
9915 PLOT x,y: DRAW 0,-7: DRAW 4,4: DRAW -4,-4: DRAW -4,4
9916 LET y=y-8
9917 LET b=PEEK a
9918 IF b=14 THEN LET a=a+4
9919 IF b=13 THEN GO TO 9922
9920 IF b=58 THEN GO TO 9940
9921 LET a=a+1: GO TO 9917
9922 IF c=20 THEN GO TO 9943
9923 LET a=a+1: GO TO 9904
9924 DRAW 12,0: DRAW 4,-8,-PI: DRAW -24,0: DRAW -4,8,-PI: DRAW 12,0: GO TO 9913
9925 DRAW 16,0: DRAW 0,-8: DRAW -32,0: DRAW 0,8: DRAW 16,0: GO TO 9913
9926 DRAW 8,-4: DRAW 8,0: DRAW -4,4: DRAW 4,-4: DRAW -4,-4: DRAW 4,4: DRAW -8,0: DRAW -8,-4: DRAW -8,4: DRAW 8,4
9927 IF b=243 THEN GO TO 9933
9928 LET a=a+1: LET f=PEEK a
9929 IF f=14 THEN LET a=a+4
9930 IF NOT f=203 THEN GO TO 9926
9931 LET a=a+1: LET f=PEEK a: PRINT AT c-2,19;CHR$ f: PRINT
9932 IF f=236 OR f=237 THEN GO TO 9935
9933 GO TO 9913
9934 CIRCLE x,y-4,4: PLOT x+4,y-4: DRAW 64,0: DRAW -4,4: DRAW 4,-4: DRAW -4,-4: PLOT x,y
9935 LET a=a+1: LET f=PEEK a
9936 IF NOT f=14 THEN GO TO 9935
9937 LET a=a+3: LET g=PEEK (a+1): LET g=g*256: LET g=g+PEEK (a)
9938 PRINT AT c-2,26;g: PRINT
9939 GO TO 9913
9940 IF c=20 THEN GO TO 9943
9941 LET a=a+1: PRINT "  ";TAB 5;CHR$ PEEK a: PRINT : LET c=c+2
9942 GO TO 9907
9943 INVERSE 1: PRINT AT 21,0;"Carregue""v""->VER""p""->PRINT": INVERSE 0
9944 IF INKEY$="v" THEN GO TO 9947
9945 IF INKEY$="p" THEN GO TO 9951
9946 GO TO 9944
9947 CLS : LET d=0: LET e=22: LET c=0: LET x=130: LET y=175
9948 IF PEEK a=58 THEN GO TO 9941
9949 IF l>9899 THEN STOP
9950 LET a=a+1: GO TO 9904
9951 PRINT AT 21,0;"
RANDOMIZE USR 65044: GO T
Q 9947

```

ESPAÇO SPECTRUM

Autor: FERNANDO D'ALMEIDA PRECES

(Parte II)

CONTINUAÇÃO DO 1.º CAPÍTULO — PROGRAMAS MONITORES

O segundo programa deste capítulo, é também dedicado à programação em código máquina. Ele está concebido para introduzir num endereço à escolha do programador, quer em código decimal quer em hexadecimal, as instruções máquina necessárias a um programa ou rotina previamente conseguido.

Mesmo para um simples ensaio em código máquina, que o leitor queira submeter ao computador, este programa monitor continua a ser útil, pois com a manobra das teclas discriminadas no pequeno programa de instruções, poderá facilmente avançar ou recuar tantos endereços quantos desejar, visualizar os códigos neles contidos, alterá-los, etc.

Até os possuidores de máquinas apenas com 16 Kbytes, apesar da extensão do Basic deste programa monitor, o podem utilizar, sem limitar o número de instruções do programa máquina a introduzir, pois o Spectrum dá-lhe a possibilidade de gravar esse programa em **fatias**, que depois poderá reagrupar após ter apagado o Basic do programa monitor.

Procurando, no entanto, reduzir ao mínimo o espaço de memória ocupado, o programa monitor está dividido em 4 secções distintas, podendo cada uma delas ser independentemente chamada pelo computador, sem carregar as restantes.

O programa «Inst» — contém o MENU com a apresentação das 3 secções monitoras e as instruções necessárias para a sua utilização. Pode modificar as cores do ecran (BORDER, PAPER e INK), ao gosto do utilizador e por selecção indica ao computador qual o programa a carregar.

O programa «1» — Introduzir as instruções máquina em código decimal ou hexa (a seleccionar) possuindo características muito úteis para os ensaios que iremos fazer ao longo deste texto.

Programa «2» — Contém 2 conversores de códigos (decimal-hexa) e (hexa-decimal), cuja utilização se destina ao estudo da linguagem máquina, pois permite, além da visualização habitual no ecran da TV, a listagem dessa conversão numa impressora (Printer).

Programa «3» — Introdução rápida de programação máquina em Hexa.

Este programa foi elaborado de forma a poupar tempo e trabalho ao utilizador pois permite a introdução de grupos de instruções de uma só vez, sendo recomendado para programas já com uma certa extensão.

Qualquer erro cometido na introdução do C.M., terá, no entanto, de ser corrigido antes de ser accionada a tecla ENTER que dá a entrada dos dados no computador, ou terá de recomeçar de novo.

```

5 BORDER 5: PAPER 5: INK 0: CLS
10 PRINT AT 10,5: FLASH 1: INK 2: " DESLIGUE O GRAVADOR "
15 PAUSE 500: FLASH 0: CLS
20 PRINT AT 10,1: PAPER 6: INK 0: " Seleccion das cores no ecran: "
25 INPUT "Moldura (Border)? ";m
30 INPUT "Papel (Paper)? ";p
35 INPUT "Tinta (Ink)? ";t
40 BORDER m: PAPER p: INK t: CLS
45 PRINT AT 3,5: PAPER 6: INK 0: " M E N U : "
50 PRINT AT 7,1: "1 - Introduutor de C.M. em          decim. ou hexa."
55 PRINT " " 2 - Conversor de decim./hexa          ou hexa/decimal."
60 PRINT " " 3 - Introduutor rapido de C.M.        em Hexadecimal."
65 PRINT " " "Selecione o Monitor (1,2 ou 3)?"
70 INPUT u: CLS
500 LET a$="O monitor faz STOP com a tecla t"
505 LET b$=" LIGUE O GRAVADOR "
510 PRINT AT 1,5: PAPER 6: INK 0: " INSTRUCCOES : "
520 IF u=2 THEN GO TO 600
530 IF u=3 THEN GO TO 700
540 PRINT " " "O monitor 1 e um programa lentomas muito util para ensaio e estu
do de programacao em codigo maquina. So admite a entrada de uma instrucao de cad
a vez sendo necessario primir ENTER antes de aintroduzir."
550 PRINT " " "As teclas de movimento (5,6,7 e8), em conjunto com tecla CAPSSH
FT, permitem recuar, avançar, 1 ou 10 enderecos de cada vez."
560 PRINT "a$
570 GO TO 800

```


600 PRINT "" O monitor 2 e um programa que converte o código Decimal em código Hexa ou este em decimal. O ecran mostra 20 instruções em cada SCROLL, sendo possível obter uma cópia destas numa Printer."

610 PRINT "" Cada instrução e convertida após primir ENTER."

620 PRINT "a\$

630 GO TO 800

700 PRINT "" O monitor 3 e um introdutor muito rápido de código máquina em Hexa. Pode introduzir todas as instruções que entender de uma só vez e estas vão sendo visualizadas à medida que são endereçadas"

710 PRINT "" Este programa e muito útil para a introdução de programação em código máquina, já devidamente testada, por não permitir recuar para emendas."

720 PRINT "a\$

800 INPUT "Após a leitura das instruções prima ENTER. "; l\$

810 CLS : PRINT AT 12,6; FLASH 1; INK 2;b\$

820 FLASH 0; PRINT "" Em carga o programa: (");u;")"

830 IF u=1 THEN LOAD "1"

840 IF u=2 THEN LOAD "2"

850 IF u=3 THEN LOAD "3"

N 100 CLS : PRINT INK 1; PAPER 5; AT 0,0; " MONITOR PARA CODIGO MAQUINA "

105 INPUT "O C.M. e em Decimal ou em Hexa? Indique 1 p/ decim. ou 2 p/hexa "; z

110 INPUT " Entre com o primeiro endereço (em decimal)"; s

120 LET s=10*INT (s/10)

130 LET x=0: LET y=0

140 GO SUB 1000

150 PRINT INK 7; PAPER 2; AT 0,0; " MONITOR PARA CODIGO MAQUINA "

160 PRINT OVER 1; FLASH 1; AT 3+2*x,3+3*y; " "

170 LET a\$=INKEY\$

180 IF CODE a\$<12 AND CODE a\$>7 THEN GO TO 600

190 IF CODE a\$=0 THEN GO TO 170

194 IF z=1 THEN GO TO 6000

196 IF z=2 THEN GO SUB 7000

210 IF LEN a\$<>2 THEN GO TO 160

220 GO SUB 500

230 IF v<0 OR v>15 THEN GO TO 170

240 LET h=v

250 LET a\$=a\$(2)

260 GO SUB 500

270 IF v<0 OR v>15 THEN GO TO 170

280 LET l=v

290 LET v=16*h+l

300 POKE s+10*x+y,v

310 PRINT OVER 1; FLASH 0; AT 3+2*x,3+3*y; " "

320 IF y=9 THEN GO SUB 1000

330 LET y=y+1

340 IF y>9 THEN LET y=0: LET x=x+1

350 IF x<10 THEN GO TO 140

360 GO TO 730

500 LET v=CODE a\$-48-7*(CODE a\$>64)-32*(CODE a\$>96)

510 RETURN

600 PRINT OVER 1; FLASH 0; AT 3+2*x,3+3*y; " "

610 GO TO 540+10*CODE a\$

620 LET y=y-1: GO TO 660

630 LET y=y+1: GO TO 700

640 LET x=x+1: GO TO 730

650 LET x=x-1: GO TO 680

660 IF y>=0 THEN GO TO 160

670 LET y=9: LET x=x-1

680 IF x>=0 THEN GO TO 140

690 LET s=s-10: CLS : GO TO 130

700 IF y<10 THEN GO TO 160

710 LET y=0: LET x=x+1

720 IF x<10 THEN GO TO 140

730 IF x=10 THEN CLS : LET s=s+10*x: GO TO 130

740 GO TO 140

1000 IF S<0 OR S>65530 THEN GO TO 110

```

1010 PRINT AT 2+2*X,0.5+10*X: PRINT TAB 2;
1020 FOR i=s+10*x TO s+10*x+9
1030 IEEE>655350#1+9: GO TO 150
1040 LET v=PEEK i: LET h=INT (v/16): LET l=v-16*h
1050 PRINT " "+CHR$ (h+48+7*(h>9))+CHR$ (l+48+7*(l>9));
1060 NEXT i
1070 RETURN
6000 REM conversao dec/hexa
6010 INPUT "Codigo maquina em decimal ";b$
6020 IF b$="t" THEN STOP
6050 LET d=VAL b$
6060 IF d>255 OR d<>INT d THEN GO TO 6010
6070 LET f=256: LET h$=""
6080 LET k=INT (d/f)
6090 GO SUB 6170
6100 LET d=d-k*f: LET f=INT (f/16)
6120 IF f<1 THEN GO TO 6140
6130 GO TO 6080
6140 IF h$(1)="0" THEN LET h$=h$+2 TO )
6160 LET a$=h$: GO TO 6210
6170 IF k<=9 THEN LET h$=h$+CHR$ (k+48)
6180 IF k>9 THEN LET h$=h$+CHR$ (k+55)
6200 RETURN
7000 INPUT "Codigo maquina em HEXA ";a$
7005 IF a$="t" THEN STOP
7010 RETURN

```

```

10 CLS : PRINT AT 2,6: PAPER 6; INK 2;" MENU DO CONVERSOR "
20 PRINT "" 1 - Decimal para Hexa."
30 PRINT "" 2 - Hexa. para Decimal."
40 PRINT ""Indique o numero que pretende."
50 INPUT c
60 IF c=2 THEN GO TO 3000
70 CLS
2000 REM conversao dec/hexa
2005 PRINT AT 10,3;"Pretende uma copia, pela Printer? Diga (s ou nao).":
INPUT p$: CLS
2010 PRINT AT 0,5; PAPER 6; INK 1;"Conversao Decim/Hexa": PRINT
2013 INPUT ; PAPER 2; INK 7;" numero em decimal. ";d$
2015 IF d$="t" THEN STOP
2020 LET d=VAL d$: LET dc=d
2035 IF d>65535 THEN INPUT ; FLASH 1; INK 2;"Fora de escala. O limite e 65535":
d$: GO TO 2020
2040 IF d<1 THEN INPUT ; FLASH 1; INK 2;"Nao introduza numeros < que 1.":d$: GO
TO 2020
2045 IF d<>INT d THEN INPUT ; FLASH 1; INK 2;"Nao introduza fraccoes.":d$: GO T
O 2020
2050 LET f=4096: LET h$=""
2060 LET y=INT (d/f)
2070 GO SUB 2200
2080 LET d=d-y*f: LET f=INT (f/16)
2090 IF f<1 THEN GO TO 2110
2100 GO TO 2060
2110 FOR n=1 TO 3: IF h$(1)="0" THEN LET h$=h$(2 TO )
2120 NEXT n
2130 PRINT TAB 5;dc,h$+" (H)"
2135 IF p$="s" THEN LPRINT TAB 5;dc,h$+" (H)"
2140 GO TO 2013
2200 IF y<=9 THEN LET h$=h$+CHR$ (y+48)
2210 IF y>9 THEN LET h$=h$+CHR$ (y+55)
2220 RETURN
3000 REM conversao hexa/decim.
3002 CLS : PRINT AT 0,5; PAPER 6; INK 1;"Conversao Hexa/Decim."
3005 PRINT AT 10,3;"Pretende uma copia, pela Printer? Diga (s ou nao).":
INPUT p$: CLS

```



```

3010 PRINT AT 0,5; PAPER 6; INK 1;"Conversao Hexa/Decim.": PRINT
3012 INPUT ; PAPER 2; INK 7;" numero em Hexa. ";h$
3015 IF h$="t" THEN STOP
3018 LET y=1: LET d=0
3020 IF LEN h$>4 THEN INPUT ; FLASH 1; INK 2;"FORA DE ESCALA - MAXIMO 4 CHR$ ";
h$
3025 LET i$=h$
3030 IF h$>="" THEN LET q$=h$(LEN h$): LET h$=h$(1 TO (LEN h$-1))
3050 GO SUB 3200: IF LEN h$>0 THEN GO TO 3030
3060 PRINT " ";i$;TAB 6;" (H) = ";d;TAB 20;" (d)"
3065 IF p$="s" THEN LPRINT TAB 5;i$;" (H) = ";d;TAB 18;" (d)"
3070 GO TO 3012
3200 LET a=CODE q$
3210 IF a<48 OR a>102 THEN GO SUB 3300
3220 IF a>57 AND a<65 THEN GO SUB 3300
3225 IF a<97 AND a>70 THEN GO SUB 3300
3230 IF a<=57 THEN LET q=a-48
3240 IF a>=65 THEN LET q=a-55
3250 IF a>=97 THEN LET q=a-87
3260 LET d=d+q*y: LET y=y*16
3270 RETURN
3300 PRINT "Num.hex c/erro...tente de novo ": PAUSE 100: RETURN

```

```

50 CLS : INPUT " Em que endereco quer iniciar o C.M.? ";x
100 INPUT " Insira o primeiro grupo de instrucoes ";a$
110 FOR n=1 TO LEN a$
115 IF a$="" THEN GOTO 100
118 IF a$="t" THEN STOP
120 LET b$=a$(1 TO 2)
130 GO SUB 300
140 POKE x,d
150 PRINT x/PEEK x
160 LET x=x+1
170 LET a$=a$(3 TO )
180 NEXT n
190 GO TO 100
300 LET d=0: LET y=1
310 FOR m=1 TO 2
320 LET d$=b$(LEN b$): LET b$=b$(1 TO (LEN b$-1))
330 LET a=CODE d$
340 IF a<=57 THEN LET q=a-48
350 IF a>=65 THEN LET q=a-55
360 IF a>=97 THEN LET q=a-87
370 LET d=d+q*y: IF y<16 THEN LET y=y*16
380 NEXT m
390 RETURN

```

(Continua no próximo número)

NOVOS PERIFÉRICOS PARA O SPECTRUM

A Firma COMPEC — Rua Santa Catarina, 715, Centro Comercial Rio (Loja O) — Telef. 314506 — 4000 PORTO, possui para entrega imediata o novo

AMPLIFICADOR DE SOM PARA O SPECTRUM

PREÇO: 3 000\$00

Terá também, a partir de 10 de Novembro próximo, os

JOYSTICKS

PREÇO: 6 000\$00

Aceitam encomendas desde que sinalizadas previamente com 30 % e remetem os JOYSTICKS e o AMPLIFICADOR DE SOM para qualquer ponto do país.

SIMULAÇÃO - EXECUTADA NO COMPUTADOR

PARTE I / OUT. 83

A simulação é um instrumento de análise poderoso, que pode ser usado pelo nosso computador, de modo a que possam ser experimentadas diferentes soluções para problemas de difícil resolução, ou que, somente com a simulação, permitirão arranjar solução (ou obter uma visão da solução possível).

A melhor forma de ter uma aproximação à simulação é começar com um exemplo concreto.

Vamos assumir a posse de um SNACK-BAR. Ao balcão existe uma pessoa, mas estamos a pensar em admitir uma segunda pessoa!

O NOSSO PROBLEMA É O SEGUINTE: ... SERÁ COMPENSADOR ADMITIR A SEGUNDA PESSOA?

Para proceder à análise do problema, recolhemos alguns dados.

O número de clientes, varia bastante com a hora do dia. O mês passado, preparamos um quadro de entradas dos clientes, que nos forneceu a seguinte panorâmica:

HORAS	N.º CLIENTES
7/ 8	30
8/ 9	15
9/10	6
10/11	3
11/12	8
12/13	25
13/14	9
14/15	8
15/16	12
16/17	12
17/18	35
18/19	22

NÓS OBSERVAMOS QUE ESTÁVAMOS A PAGAR UMA PENALIZAÇÃO, PELO FACTO DE NÃO TERMOS UM SEGUNDO EMPREGADO AO BALCÃO.

Nos casos em que o cliente teria de esperar muito tempo para ser atendido, ele sairia para procurar outro snack-bar.

Observámos ainda que, em média, as pessoas que entravam no bar ficavam ou saíam, em função do número de pessoas que estavam a ser servidas ao balcão.

RESULTADOS DAS NOSSAS OBSERVAÇÕES:

N.º pessoas ao balcão	% Entra e fica	% Entra e sai
0	0	0
1- 3	15	5
4- 6	25	15
7-10	60	35
11-15	80	50

O tempo médio que uma pessoa espera para ser atendida é de 4 minutos e a despesa média de Esc. 57\$50.

O salário semanal do novo empregado será de 2 000\$00. Assumindo que o novo vendedor trabalha

continuamente enquanto o bar está aberto, qual a decisão que devo tomar?

Este problema é típico na área dos assuntos comerciais e industriais. Caracteriza-se pelos dados acumulados a partir de observações e acontecimentos imprevisíveis (a que horas vai chegar um cliente?; irá o cliente impacientar-se pela demora e sair?).

Uma das técnicas que nos irá ajudar a tomar a decisão é a de deixar que o nosso computador «imite» o snack-bar; deixá-lo «jogar», ou seja, «gerar clientes» em tempos aleatórios.

Estes clientes entram no bar e, na base do número de pessoas que ali estiverem, decidirão ficar ou sair.

O computador segue todos os dados, calcula os lucros e as perdas. Irá simular toda a actividade de um dia e apresentará os resultados correspondentes.

Mas, atenção! O computador pode gerar dados para um dia «não típico».

Estes clientes entram no bar na base do habitual, ou seja, decidem se ficam para ser atendidos ou, pelo contrário, podem ir procurar outro local. O computador levará todos estes valores em conta mas, devido ao facto de os dados serem aleatórios, o programa irá funcionar para vários dias simulados e obter o resultado médio.

Em seguida, iremos ver algumas ideias sob o ponto de vista da matemática. Especificamente, iremos repensar algumas noções sobre «como consegue o computador imitar um acontecimento imprevisível?».

Considere-se, por exemplo, o cliente que entra e encontra quatro pessoas a ocupar o balcão. De acordo com a tabela anteriormente criada, o cliente ocupa 25 % do tempo na fase de ser atendido e 75 % do tempo será ocupado ao balcão a comer ou beber. Como obter do computador o mesmo tipo de decisão, como se fosse o cliente?

O método mais fácil é usar o gerador de números aleatórios. Faça um apelo a RND, que é a forma de obter um número aleatório entre 0 (incluído) e 1 (excluído). Sabendo isto, nós queremos usar a ideia de que, conhecendo o instante em que o valor é RND é superior a 25 %, então isso será equivalente ao facto de que o cliente entrou e ficou ao balcão. Em todas as outras vezes, será equivalente ao caso de um cliente que entrou e saiu de novo.



$x \leq .25$ equivale a cliente que entra e sai

$x > .25$ equivale a cliente que entra e fica

(Continua no próximo número)

NOVOS PROGRAMAS PARA O SPECTRUM

● EDUCAÇÃO: em português

LÓGICA: TEXTO AUXILIAR DA LÓGICA MATEMÁTICA

TRANSFORMAÇÕES GEOMÉTRICAS: Contém 4 subprogramas; Translação; Rotação; Simetria; Homotetia

ESTES PROGRAMAS ESTÃO DE ACORDO COM O PROGRAMA DE ENSINO OFICIAL DA MATEMÁTICA.

● AVENTURAS: em inglês

FEUDAL OVERLORD: o jogador simula a posse de uma terra e a sua administração de acordo com as regras do feudalismo.

SMUGGLERS COVE: história de piratas.

DEVILS OF THE DEEP: terá como missão explorar a cidade perdida da ATLÂNTIDA.

TRAIN: será o chefe de operações do tráfego ferroviário.

HEATHROW — AIR TRAFIC CONTROL: irá controlar o tráfego aéreo no aeroporto de HEATHROW.

● JOGOS

POOL — (BUG BYTE) jogo de bilhar para 16 k ou 48 k.

POOL — (CDS) jogo de snooker (bilhar) 16 k ou 48 k

● ASTROLOGIA

STARGAZER SECRETS: A partir da data de nascimento faz o traçado da carta astrológica e análise do carácter (em inglês).

(Está todo em BASIC e pode ser traduzido ou adaptado para português).

● UTILITÁRIOS

48 K MELBOURNE DRAW: Para criar gráficos; reduzir ou aumentar imagens; examinar secções do ecran; gravar figuras em cassete; criar e definir UGDS.

SOFKIT: renumeração; eliminar grupos de linhas; passar de minúsculas para maiúsculas ou vice-versa; pesquisa de qualquer carácter, frase ou num. ou instrução; trace.

DEVPAC: ASSEMBLER/EDITOR DISASSEMBLER

● LINGUAGENS

PASCAL (4T DA HISOFT... NOVA VERSÃO)

preços previstos:

jogos, aventuras e astrologia (300\$00)

educacional (600\$00)

utilitários (1 000\$00... Melb. draw)

(600\$00... Sofkit e Devpac)

Linguagens (1 200\$00)

O desconto de 20 % continua a ser válido para os sócios do clube.



CLUBE Z₈₀

INSCRIÇÃO COMO ASSOCIADO

O **CLUBE Z₈₀** está aberto a todos os utilizadores de microcomputadores.

A intenção de associar os entusiastas das micro-máquinas, é exclusivamente a de permitir:

- 1 — PUBLICAÇÃO DE UM JORNAL MENSAL, onde sejam publicados programas de uso geral ou específico como no caso da educação.
- 2 — PROMOVER TROCAS DE PROGRAMAS, e trocas de experiências; tanto no caso do Software (programação), como no caso do Hardware (electrónica).
- 3 — PROMOVER DESCONTOS NA AQUISIÇÃO DE PROGRAMAS.
- 4 — LANÇAR CURSOS DE PROGRAMAÇÃO EM BASIC — PASCAL OU OUTRAS LINGUAGENS E DIVULGAR O USO DE LINGUAGEM MÁQUINA.

NOME _____

IDADE _____ COMPUTADOR TIPO _____

PROFISSÃO _____

ENDEREÇO _____

TELEF. _____

ASSINATURA ANUAL — Esc. 1 500\$00 ☐

ASSINATURA SEMESTRAL — Esc. 750\$00 ☐

CHEQUE OU VALE DO CORREIO

N.º _____

BANCO _____

DATA ____/____/____

JÁ SÓCIO ☐

NOVO SÓCIO ☐ → A partir do mês de _____ (inclusive)